@echo off
chcp 65001 >nul
setlocal enabledelayedexpansion

REM 应用名称
set APP_NAME=boot

REM Java虚拟机启动参数
set JAVA_OPTS=-Xms512m -Xmx1024m

REM 获取应用的目录
set SCRIPT_DIR=%~dp0
set APP_HOME=%SCRIPT_DIR:~0,-5%

REM 配置文件目录
set CONF_DIR=%APP_HOME%\classes

REM lib目录
set LIB_DIR=%APP_HOME%\lib

REM 使用通配符方式设置CLASSPATH，避免命令行过长
set CLASSPATH=%CONF_DIR%;%LIB_DIR%\*

REM 解析传入的参数
set PROFILE_ARG=
set OTHER_ARGS=
:parse
if "%1"=="" goto endparse
if "%1"=="--spring.profiles.active" (
    set PROFILE_ARG=--spring.profiles.active=%2
    shift
    shift
    goto parse
)
if "%1"=="-spring.profiles.active" (
    set PROFILE_ARG=--spring.profiles.active=%2
    shift
    shift
    goto parse
)
set OTHER_ARGS=%OTHER_ARGS% %1
shift
goto parse
:endparse

REM 确保日志目录存在
if not exist logs mkdir logs

REM 启动应用
java %JAVA_OPTS% -cp "%CLASSPATH%" io.geekidea.boot.SpringBootPlusApplication %PROFILE_ARG% %OTHER_ARGS%
echo %APP_NAME% started successfully!